JOURNAL OF RESEARCH of the National Bureau of Standards—C. Engineering and Instrumentation
Vol. 72C, No. 2, April–June 1968

Analysis of n-Degree Elliptical Elastic Rings of Non-uniform Cross Section

R. A. Mitchell

Institute for Basic Standards, National Bureau of Standards, Washington, D.C. 20234

(February 15, 1968)

A curved bar structural analysis is developed for an n-degree elliptical elastic ring with sinusoidally varying cross-section dimensions, where the n-degree ellipse is defined by

$$\left(\frac{x}{a}\right)^n+\left(\frac{y}{b}\right)^n=1, \qquad n \geqslant 2.0.$$

A sequential grid constrained optimization method is used to search for a minimum weight design in six dimensional shape parameter space. Numerical results indicate that rings of this shape can be designed to have considerably less weight and greater flexibility than comparable circular rings with uniform cross-section dimensions.

Key Words: Curved bar analysis, elasticity, force transducer, n-degree elliptical ring, nonlinear programming, optimization, proving ring, structural design, theory.

1. Introduction

Elastic ring force transducers (such as proving rings, load rings, ring dynamometers, etc.), for a wide range of load carrying capacities have most frequently been made circular in shape with uniform rectangular cross section. This shape has imposed limitations on the usefulness of this type device, particularly for higher capacities (i.e., greater than 100,000 lbf), due to overall size, gross weight, limited deflection range, and size effect during heat treatment. For the same force resisting capacity, rings of more complex shapes can be designed to have greater deflection ranges, less weight, and smaller maximum dimensions. This paper describes methods of analysis of a broad geometric class of complex shaped rings which result in improved deflection and weight characteristics.

Structural analysis and design optimization methods are developed for an n-degree elliptical elastic ring with sinusoidally varying cross-section dimensions. The ring is analyzed as a thick curved bar, assuming small deflections, plane strain, Winkler-Bach bending stress distribution, uniform normal stress distribution, and parabolic shear stress distribution [1].[1] A sequential grid constrained optimization method is used to search for the minimum weight ring design that will satisfy the specific force capacity, deflection, and dimensional requirements of a force transducer.

[1] Figures in brackets indicate the literature references on page 150.

List of Symbols

x, y rectangular coordinates of ellipse,
r, θ polar coordinates of ellipse,
n degree of ellipse,
a, b semimajor and semiminor axes of ellipse,
h ring cross-section width,
t ring cross-section thickness normal to tangent of ellipse,
c average of maximum and minimum ring width,
d average of maximum and minimum ring thickness,
e dimensionless parameter of ring width variation,
f dimensionless parameter of ring thickness variation,
s arc length variable measured along ellipse,
ψ counterclockwise angle from radius vector to tangent line,
L total load applied to ring,
P $=L/2$, load applied to one quadrant of ring,
M moment,
N normal force,
V shearing force,
σ normal stress,
τ shearing stress,
ϵ strain,
k curvature of ellipse,
β relative rotation of differential segment ends due to N,
Θ curved bar rotation,
δ curved bar deflection,
Δ total ring deflection,
U strain energy of curved bar,
u strain energy per unit volume,
ν Poisson's ratio,
E modulus of elasticity,
g integral defined by eq (19),
S design maximum stress,
λ scale factor,
W ring weight,
γ material weight per unit volume,
g_i constraint function,
b_i constraint bound,
x_j variable of the optimization problem,
z objective function.

2. Structural Analysis

An elastic ring force transducer is loaded by two forces acting in opposite directions along a diameter, as shown in figure 1. Corresponding ring deflections are measured and related to applied forces by a calibration factor.[2] For a ring of the geometric class represented in figure 1, the locus of centroids of ring cross sections is an n-degree ellipse defined by the equation

$$\left(\frac{x}{a}\right)^n + \left(\frac{y}{b}\right)^n = 1, \qquad n \geqslant 2.0, \tag{1}$$

[2] Wilson, Tate, and Borkowski [2] have described the uniform cross section, circular force transducer, its calibration, and performance under various conditions of use.



FIGURE 1. 1,000,000 *lbf capacity* n-*degree elliptical ring with sinusoidally varying thickness.*

or in polar coordinate form

$$r = ab(b^n \cos^n \theta + a^n \sin^n \theta)^{-1/n}. \tag{2}$$

Several n-degree ellipses are plotted in figure 2. Ring rectangular cross-section width h and thickness t are given by

$$h = c(1 - e \cos 2\theta), \tag{3}$$

and

$$t = d(1 - f \cos 2\theta), \tag{4}$$

in which c and d are positive parameters and e and f are parameters of absolute value less than unity. Using these equations, the shape of a ring within this geometric class can be specified by the seven parameters a, b, c, d, e, f, and n. If $a = b$, $n = 2$, and $e = f = 0$, the ring shape is circular with uniform rectangular cross section.

One quadrant of the loaded ring is shown in figure 3. Due to symmetry there is no rotation at $\theta = 0$ or at $\theta = \pi/2$, no shear force V at $\theta = 0$, and no normal force N at $\theta = \pi/2$. Expressions for the



FIGURE 2. n-*degree elliptical curves.*



FIGURE 3. *Loads acting on one quadrant of a ring.*

resultant moment M and forces N and V acting on a typical normal cross section are derived by statics using formulas from calculus [3] as follows:

$$\frac{d\theta}{ds}=\frac{\sin\psi}{r}=\frac{1}{\sqrt{r^2+\left(\frac{dr}{d\theta}\right)^2}}, \tag{5}$$

$$\frac{dr}{ds}=\cos\psi=\frac{dr}{d\theta}\frac{d\theta}{ds}, \tag{6}$$

$$M=M_{\pi/2}-Pr\cos\theta=PM', \tag{7}$$

$$N=P\sin(\theta+\psi)=P\left(\sin\theta\frac{dr}{ds}+r\cos\theta\frac{d\theta}{ds}\right)=PN', \tag{8}$$

and

$$V=-P\cos(\theta+\psi)=P\left(r\sin\theta\frac{d\theta}{ds}-\cos\theta\frac{dr}{ds}\right)=PV'. \tag{9}$$

Differentiation of eq (2) gives

$$\frac{dr}{d\theta}=r\frac{(b^n\cos^{(n-1)}\theta\sin\theta-a^n\sin^{(n-1)}\theta\cos\theta)}{(b^n\cos^n\theta+a^n\sin^n\theta)}. \tag{10}$$

The assumption that ring cross sections remain plane leads to the Winkler-Bach bending stress formula, which has been written by Seely and Smith [1] in the following form, valid for a straight or curved bar:

$$\sigma_M=\frac{M}{h}\left[\frac{k}{t}+\frac{y}{(1+ky)\int_t\frac{y^2}{(1+ky)}dy}\right]. \tag{11}$$

The variable y is the distance from the cross-section centroid to a point in the cross section, positive in the outward direction. The curvature k, the inverse of the radius of curvature, is given by [3]

$$k=\frac{r^2+2\left(\frac{dr}{d\theta}\right)^2-r\frac{d^2r}{d\theta^2}}{\left[r^2+\left(\frac{dr}{d\theta}\right)^2\right]^{3/2}}. \tag{12}$$

Differentiation of eq (10) gives the required second derivative

$$\begin{aligned}\frac{d^2r}{d\theta^2}=&(n+1)ab[a^n\sin^{(n-1)}\theta\cos\theta-b^n\cos^{(n-1)}\theta\sin\theta]^2[a^n\sin^n\theta+b^n\cos^n\theta]^{-(2+1/n)}\\&+a^{(n+1)}b[\sin^n\theta-(n-1)\sin^{(n-2)}\theta\cos^2\theta][a^n\sin^n\theta+b^n\cos^n\theta]^{-(1+1/n)}\\&+ab^{(n+1)}[\cos^n\theta-(n-1)\cos^{(n-2)}\theta\sin^2\theta][a^n\sin^n\theta+b^n\cos^n\theta]^{-(1+1/n)}.\end{aligned} \tag{13}$$

It is assumed that stress due to the normal force N is uniformly distributed over the cross section, that is

$$\sigma_N=\frac{N}{ht}. \tag{14}$$

It is assumed that shear stress is distributed parabolically over the rectangular cross section according to the formula

$$\tau_V = \frac{V}{ht}\left(\frac{3}{2} - 6\,\frac{y^2}{t^2}\right). \tag{15}$$

Castigliano's strain energy theorem [1] is used to determine the ring bending moment and deflection at $\theta = \pi/2$. The theorem stated symbolically as applied to this problem is

$$\Theta_{\pi/2} = \frac{\partial U}{\partial M_{\pi/2}}, \tag{16}$$

and

$$\delta_{\pi/2} = \frac{\partial U}{\partial P}, \tag{17}$$

in which U = total strain energy of the loaded bar,
$\Theta_{\pi/2}$ = bar rotation at $\pi/2$,
and $\delta_{\pi/2}$ = bar deflection at $\pi/2$ in direction of P.

The strain energy per unit volume due to bending stress is

$$u_M = \frac{1}{2}\,\sigma_M \epsilon_M = \frac{(1-\nu^2)\sigma_M^2}{2E}, \tag{18}$$

in which ϵ_M = unit plane strain,
ν = Poisson's ratio,
and E = modulus of elasticity.

Substituting eq (11) into (18) and integrating over a differential segment gives the corresponding strain energy of the segment

$$dU_M = \frac{(1-\nu^2)M^2}{2Eh}\left\{\int_t \left[\frac{k}{t} + \frac{y}{(1+ky)\int_t \frac{y^2}{(1+ky)}\,dy}\right]^2 dy\right\} ds = \frac{(1-\nu^2)M^2}{2Eh}\,g\,ds. \tag{19}$$

The strain energy per unit volume due to the normal force N is

$$u_N = \frac{1}{2}\,\sigma_N \epsilon_N = \frac{1}{2}\left(\frac{N}{ht}\right)\left(\frac{N}{htE}\right) = \frac{N^2}{2Eh^2t^2}, \tag{20}$$

and the strain energy of a differential segment is

$$dU_N = \frac{N^2}{2Eht}\,ds. \tag{21}$$

The strain energy per unit volume due to the shear force V is

$$u_V = \frac{(1+\nu)\tau^2}{E} = \frac{(1+\nu)V^2}{Eh^2t^2}\left(\frac{3}{2} - 6\,\frac{y^2}{t^2}\right)^2, \tag{22}$$

and the strain energy of a differential segment is

$$dU_V = \frac{6}{5}\left(\frac{1+\nu}{E}\right)\frac{V^2}{ht}\,ds. \tag{23}$$

The normal force N acting on a differential segment causes rotation of one end cross section relative to the other, due to the difference in arc length from inside radius to outside radius. The mean change in segment arc length is

$$\frac{N}{Eht}\,ds=\left(\frac{1}{k}\right)\beta, \tag{24}$$

in which β = rotation due to N.

The bending moment M acting during this rotation contributes strain energy to the segment equal to

$$dU_{MN}=\frac{kMNds}{Eht}. \tag{25}$$

Adding eqs (19), (21), (23), and (25) gives the total strain energy of a differential segment

$$dU=\frac{1}{E}\left[\frac{N^2}{2ht}+\frac{6}{5}(1+\nu)\frac{V^2}{ht}+\frac{1}{2}(1-\nu^2)\frac{M^2g}{h}+\frac{MNk}{ht}\right]ds. \tag{26}$$

Applying eq (16) and the boundary condition of no rotation at $\theta=\pi/2$ gives

$$\frac{\partial U}{\partial M_{\pi/2}}=\frac{1}{E}\int_0^{\pi/2}\left[\frac{N}{ht}\frac{\partial N}{\partial M_{\pi/2}}+\frac{12}{5}(1+\nu)\frac{V}{ht}\frac{\partial V}{\partial M_{\pi/2}}+(1-\nu^2)\frac{gM}{h}\frac{\partial M}{\partial M_{\pi/2}}+\frac{kM}{ht}\frac{\partial N}{\partial M_{\pi/2}}+\frac{kN}{ht}\frac{\partial M}{\partial M_{\pi/2}}\right]ds$$

$$=\frac{1}{E}\int_0^{\pi/2}\left[\frac{(1-\nu^2)gM}{h}+\frac{kN}{ht}\right]ds=0. \tag{27}$$

Substitution of eqs (7) and (8) into (27) gives

$$M_{\pi/2}=P\left[\frac{\int\frac{(1-\nu^2)gr\cos\theta}{h}ds-\int\frac{k\sin\theta}{ht}dr-\int\frac{kr\cos\theta}{ht}d\theta}{\int\frac{(1-\nu^2)g}{h}ds}\right]. \tag{28}$$

Applying eq (17) to (26) gives

$$\delta_{\pi/2}=\frac{\partial U}{\partial P}=\frac{1}{E}\int_0^{\pi/2}\left[\frac{N}{ht}\frac{\partial N}{\partial P}+\frac{12(1+\nu)V}{5ht}\frac{\partial V}{\partial P}+(1-\nu^2)\frac{gM}{h}\frac{\partial M}{\partial P}+\frac{kM}{ht}\frac{\partial N}{\partial P}+\frac{kN}{ht}\frac{\partial M}{\partial P}\right]ds$$

$$=\frac{P}{E}\int_0^{\pi/2}\frac{1}{ht}\left[N'^2+\frac{12}{5}(1+\nu)V'^2-(1-\nu^2)gtr\cos\theta M'+kM'N'-kr\cos\theta N'\right]ds. \tag{29}$$

The tangential stress on the inner surface is provisionally assumed to be the limiting stress that determines the force resisting capacity of the ring. The possibility of capacity being limited by some other component of stress or combined stress should be checked in any final design analysis. Substituting $-t/2$ for y outside the integral in eq (11) and adding to eq (14) gives the tangential stress on the inner surface of the ring

$$\sigma_T=P\left[\frac{M'k}{ht}-\frac{M't}{h(2-kt)\int_t\frac{y^2}{(1+ky)}dy}+\frac{N'}{ht}\right]. \tag{30}$$

For a circular ring of uniform cross section the maximum value of σ_T occurs at $\theta = \pi/2$. But for the more general geometry considered here the maximum may occur at $\theta = 0$ or at some point in between.

The structural analysis thus far has been developed for one quadrant of the ring. For the total ring, figure 1, the applied force is

$$L = 2P, \tag{31}$$

and the deflection at the point of load application is

$$\Delta = 2\delta_{\pi/2}. \tag{32}$$

3. Design Analysis

The design analysis attempts to find the dimensions of the minimum weight ring that will satisfy the arbitrary force capacity, deflection, and dimensional requirements of a particular force transducer. The deflection range of the ring must be compatible with the deflection sensing device, and the inside dimensions of the ring must be sufficient to accommodate the sensor. Outside dimensions of the ring must not interfere with its intended use. The thickness of the ring must permit the heat treatment required for the ring material. All trial solutions are scaled up or down to coincide with the force capacity requirement.

3.1. Scale Factors

The procedure adopted here for meeting the force capacity requirement is to scale all length dimensions in the r-θ plane by a factor λ that will make the maximum stress (the maximum value of σ_T for the entire ring) equal a prescribed maximum design stress S. The cross-section width h is not scaled. This gives the specialized equation for the maximum tangential stress on the inner surface of the scaled ring

$$S = \frac{L_\lambda}{2}\left(\frac{1}{\lambda}\right)\left[\frac{M'k}{ht} - \frac{M't}{h(2-kt)\int_t \frac{y^2}{(1+ky)}\,dy} + \frac{N'}{ht}\right]_{\text{MAX.}} \tag{33}$$

$$= \frac{L_\lambda F}{2\lambda},$$

in which S = design maximum stress,

L_λ = design capacity,

and F = maximum value of bracketed function.

This gives the scale factor

$$\lambda = \frac{L_\lambda F}{2S}. \tag{34}$$

Scaled ring dimensions, λ subscripted, are

$$r_\lambda = \lambda r, \tag{35}$$

and

$$t_\lambda = \lambda t. \tag{36}$$

Scaled forces and moment are

$$L_\lambda = \lambda L, \tag{37}$$

$$N_\lambda = \lambda N, \tag{38}$$

$$V_\lambda = \lambda V, \tag{39}$$

and

$$M_\lambda = \lambda^2 M. \tag{40}$$

Scaled ring deflection is

$$\Delta_\lambda = \lambda \Delta. \tag{41}$$

Scaled ring weight is

$$W_\lambda = 4\gamma\lambda^2 \int_0^{\pi/2} htds = \lambda^2 W \tag{42}$$

in which γ = material weight per unit volume.

3.2. Thin Circular Ring

By making the appropriate simplifying assumptions, the thick ring equations developed above can be reduced to relatively simple thin circular ring equations. The thin circular ring equations, although they may be in error by several percent, can be used to determine a reasonable range of ring proportions for beginning an optimum search procedure. The equations also give a useful indication of the comparative efficiencies of different ring materials.

A thin ring is here defined as one in which the ratio of thickness to radius of curvature is so small that, with negligible error: (1) the stress distribution due to bending can be assumed linear over a cross section, and (2) deflection due to shear V and normal force N can be neglected. By assuming zero effective curvature and no deflection due to V or N, the following thin ring equations for moment, maximum bending stress, and deflection at the point of load application can be obtained from eqs (28) through (32):

$$M = \frac{Lr}{\pi}, \tag{43}$$

$$\sigma_T = \frac{6M}{ht^2}, \tag{44}$$

and

$$\Delta = \left(3\pi - \frac{24}{\pi}\right) \frac{(1-\nu^2)Lr^3}{Eht^3}. \tag{45}$$

These equations, solved simultaneously, give

$$t = \sqrt[3]{\frac{72EL^2\Delta}{(\pi^4 - 8\pi^2)(1-\nu^2)\sigma_T^3 h^2}}, \tag{46}$$

and

$$r = \frac{\pi \sigma_T h t^2}{6L}. \tag{47}$$

The weight of a thin circular ring is

$$W = 2\pi rht\gamma = \frac{24E\gamma L\Delta}{(\pi^2-8)(1-\nu^2)\sigma_T^2}. \tag{48}$$

Note that the weight, for a particular $L\Delta$, is a function of the material properties E, ν, σ_T, and γ. Equation (48) suggests the importance of maximum permissible design stress in a comparison of different ring materials where weight is a significant factor.

3.3. Optimization Method

The optimization problem considered here is of the general nonlinear programming class. The goal of this type problem is to determine the set of values of the variables $x_1, \ldots, x_p$ which satisfies the constraint conditions

$$g_i(x_1, \ldots, x_p)\{\leqslant, =, \geqslant\}b_i,\ i=1, \ldots, m, \tag{49}$$

and minimizes the objective function

$$z = f(x_1, \ldots, x_p). \tag{50}$$

In the present case the p variables are the six independent ring shape parameters a, c, d, e, f, and n. The structural analysis is first done for a ring of unit value of mean radius, that is

$$\frac{a+b}{2} = 1. \tag{51}$$

Thus the value of the parameter a determines the value of b. The thickness parameter d is, for the unit size ring, the ratio of the mean of the extreme values of thickness to mean radius. After the structural equations are solved for the unit mean radius ring (using Simpson numerical integration), the results are scaled to full size using the scale factor λ. The width parameter c is not scaled during the entire analysis, and the shape parameters e, f, and n are nondimensional and are not scaled.

The m constraint functions g_i and the corresponding constraint bounds b_i are as follows:

(1) Ring deflection

$$g_1 = \Delta_\lambda \geqslant b_1, \tag{52}$$

(2) Maximum thickness of ring cross section

$$g_2 = \lambda d(1+|f|) \leqslant b_2, \tag{53}$$

(3) Outside width of ring

$$g_3 = \lambda(2a+d-df) \leqslant b_3, \tag{54}$$

(4) Outside height of ring

$$g_4 = \lambda(4-2a+d+df) \leqslant b_4, \tag{55}$$

(5) Maximum width of ring cross section

$$g_5 = c(1+|e|) \leq b_5, \tag{56}$$

(6) Inside clear height of ring

$$g_6 = \lambda(4-2a-d-df) \geq b_6. \tag{57}$$

The constraint bounds b_i are the limits on ring characteristics imposed by use, fabrication, or material requirements.

The objective function z to be minimized is the ring weight W_λ, eq (42).

A sequential grid method was used with reasonable success to search for minimum weight solutions. The method requires several computer runs of moderate duration. Between computer runs changes are made in the input data based on the experience of previous runs. The "Elastic Ring Optimization Program" listed in the appendix was used for this search procedure.

For the first computer run, one or more values of each of the six variables x_j are chosen as input data. If two or more values of a variable are chosen, they are widely spaced, but within a reasonable range for that variable. By nested do loops the computer generates all combinations of the variables. Each combination is a solution point in the six dimensional shape parameter space. The ring weight and the six constraint function values are computed and printed for each solution point. The best solution point for a computer run is the one with the least weight that also falls within all six constraint bounds.

Subsequent computer runs are made using as input data variable values distributed about the best known solution point. The range of values of any variable should include the value of that variable for the best known solution point. Any results of previous computer runs may be useful as a guide for selecting input data. The process is repeated until a solution point with satisfactorily small weight is found, or until results indicate that a satisfactory approximation of a minimum weight solution has been found.

Computer time required for each solution point is approximately one-half second on a UNIVAC 1108 computer. Obviously there are practical limits to the number of variable combinations one would want to generate in one computer run.

A second computer program (the "Detailed Ring Analysis Program" listed in the appendix) can be used to compute scaled dimensions, forces, moments, and stresses for a particular set of input shape parameters. The program prints r_λ, t_λ, h, k, N_λ, V_λ, M_λ, σ_T, and the value of the integral in eq (11) for each incremental value of θ. These results can be used with eqs (11), (14), and (15) to determine whether ring capacity is limited by a stress condition other than the maximum value of σ_T.

The following numerical example demonstrates the optimization method.

3.4. Numerical Example

The problem is to determine, for a ring of 1,000,000 lbf capacity and for the material constants and constraint bounds listed below:

(1) The approximate weight of a circular ring of uniform rectangular cross section.

(2) The dimensions of an approximately minimum weight n-degree elliptical ring of constant cross section width and sinusoidally varying thickness.

(3) The dimensions of an approximately minimum weight n-degree elliptical ring of sinusoidally varying cross section width and thickness.

The material constants are: $S=150{,}000$ lb/in^2, $E=30{,}000{,}000$ lb/in^2, $\nu=0.3$, and $\gamma=0.29$ lb/in^3. The constraint bounds are: $b_1=0.25$ in, $b_2=4.5$ in, $b_3=40.0$ in, $b_4=40.0$ in, $b_5=14.0$ in, $b_6=9.0$ in.

a. Solution 1

Thin circular ring equations were used to compute a rough approximation of ring weight and dimensions. Substitution of L, E, γ, ν, S for σ_T, and b_1 for Δ into eq (48) gives an approximate ring weight of 1365 lb. Substitution of the same values, along with b_5 for h, into eqs (46) and (47) give $t=3.66$ in and $r=14.73$ in. The ratio of thickness to radius is 0.248. The more accurate thick curved bar equations were used for all subsequent computations.

Computer run No. 1 of the sequential grid search included nine uniform cross-section circular rings with widths of 10, 12, and 14 in and thickness to radius ratios of 0.20, 0.25, and 0.30. For these nine cases the ratios of weight to deflection differed by less than two percent, a result that might have been anticipated from eq (48). The two cases with characteristics nearest the constraint bounds had deflections of 0.234 and 0.277 in and weights of 1327 and 1587 lb respectively. Linear interpolation between these two cases gives an estimated weight of 1424 lb for a ring that would deflect 0.250 in.

b. Solution 2

A sequence of five computer runs were used to seek the dimensions of a minimum weight, constant width ring. The input shape parameter data and the resulting weight for the best point of each run are given in table 1. The best point parameter values for each run are underlined. The weight of 553 lb for the fifth run is about 39 percent of the estimated weight of 1424 lb for a uniform cross-section circular ring. The 553 lb ring was accepted as a satisfactory approximation of the minimum weight constant width ring. The ring shape is shown in figure 1. Critical ring dimensions and a deflection of 0.257 in all fall within the constraint bounds.

Plots of σ_T versus θ for the 553 lb ring and for a 1587 lb uniform circular ring ($c=14.0$ in, $d=0.25$ in, $\Delta=0.277$ in in computer run No. 1) are shown in figure 4. The high stress level over a greater portion of the 553 lb ring is an indication of more efficient use of material in this more complex ring shape.

c. Solution 3

Computer runs 6 and 7 included many cases with variation of cross-section width. But for both runs the best points were constant width cases. The seventh run included a variable width case with parameter values $c=13.6$ in, $e=0.025$, and $f=0.45$, and a resulting weight of 539.18 lb which

TABLE 1. *Sequential grid search data for numerical example*

Computer run No.	Independent ring shape parameters[a]						Weight for best point
	a	*c*	*d*	*e*	*f*	*n*	
	Inch	*Inch*	*Inch*				*Lb*
1	1. 1.2	10. 12. 14.	.2 .25 .3	0.	0. .2	2. 3. 4.	802.00
2	1.1 1.3	13. 14.	.25 .3 .35	0.	.1 .2 .3	2.5 3. 3.5	828.42
3	1.15 1.25	13.5 14.	.28 .3 .32	0.	.2 .3 .4	3. 3.5 4.	639.84
4	1.23 1.25 1.27	13.4 13.7	.31 .33	0.	.3 .35 .4	4. 4.5 5.	572.90
5	1.22 1.23 1.24	13.6 13.8	.305 .31 .315	0.	.4 .45	4.5 5. 5.5	553.06
6	1.21 1.22	12. 12.7	.3 .305	−.1 0. .1	.45 .5	5.5 6.	564.55
7	1.21	12.7 13. 13.3 13.6 14.	.305	−.025 0. .025 .05 .075 .1	.4 .45 .5	6.	539.08

[a] Best point values for each computer run are underlined.



FIGURE 4. *Tangential stress on inner surface of nonuniform n-degree elliptical ring and uniform circular ring.*

292–251 O–68—3

is only slightly greater than the best point weight. Additional searching might result in a slight weight reduction due to width variation. But the weight savings would probably not be enough to justify the additional complexity of shape.

4. Summary

A structural analysis method has been developed for an n-degree elliptical elastic ring with sinusoidally varying cross-section dimensions. A sequential grid constrained optimization method has been used to search for the minimum weight ring of this geometric class that would be suitable as the load supporting element of a force transducer.

Limited numerical results were obtained for one-million pound force capacity n-degree elliptical rings with nonuniform cross-section dimensions. The results indicate that such a ring with constant cross-section width and sinusoidally varying thickness can be designed to weigh as little as 39 percent of the weight of a comparable uniform cross section, circular ring. Outside dimensions of the more complex ring are also significantly less than those of the uniform, circular ring. The results for an n-degree elliptical ring with sinusoidally varying cross-section width and thickness suggest that the weight savings due to variation of width may not be great enough to justify that additional complexity of shape.

The Elastic Ring Optimization Program (given in appendix) can be used directly for design optimization by the sequential grid method described in this paper; or the program could be used in combination with some other optimization procedure. In either case, the Detailed Ring Analysis Program (see appendix) or another comprehensive stress analysis should be used for a detailed examination of stresses for the entire ring.

The analysis was coded in FORTRAN V for the UNIVAC 1108 computer by R. M. Slesser of the National Bureau of Standards Center for Computer Sciences and Technology.

5. References

[1] F. B. Seely and J. O. Smith, Advanced Mechanics of Materials, 2d ed. (John Wiley & Sons, Inc., (1952)).
[2] B. L. Wilson, D. R. Tate, and G. Borkowski, Proving Rings for Calibrating Testing Machines, Circular of the National Bureau of Standards C454 (1946).
[3] E. S. Smith, M. Salkover, and H. K. Justice, Unified Calculus (John Wiley & Sons, Inc. (1947)).

6. Appendix

The two FORTRAN V computer programs used in the elastic ring design analysis are listed below. The Elastic Ring Optimization Program is used to search for the optimum set of shape parameters a, c, d, e, f, and n. The Detailed Ring Analysis Program is used to compute scaled dimensions, forces, moments, and stresses for a particular set of input shape parameters. Table A1 lists the computer symbols used to represent mathematical symbols appearing in the equations of the text. Sample input data is included at the end of each program.

TABLE A1. *Corresponding symbols in text, computer programs, and computer output*

Text	Programs	Output
a	C1	A
c	C3	C
d	C5	D
e	C7	E
f	C8	F
n	XN	N
L_λ	XLLAM	L LAMBDA
S	S	S
E	E	E
ν	XNHU	NU
γ	GAMMA	GAMMA
h	B(IJ)	L
t	T(IJ)	T
r	R(IJ)	R
θ	THETA(IJ)	THETA
k	RHO(IJ)	K
g	G(IJ)	
M'	PM(IJ)	
N'	PN(IJ)	
V'	PV(IJ)	
F	DMAX	
λ	XLAM	LAMBDA
W_λ	WLAM	WEIGHT
g_1	DELAM	G1
g_2	TCON4	G2
g_3	TCON5	G3
g_4	TCON6	G4
g_5	TCON7	G5
g_6	TCON8	G6
y	Y	
λN	XLN(I)	LAMN
λV	XLV(I)	LAMV
$\lambda^2 M$	XLM(I)	LAM**2*M
λr	RLAM	LAMR
λt	TLAM	LAMT
σ_T	SIGTL(I)	SIGMA
θ for max. σ_T	JMAX	TM
W_λ/Δ_λ	RATIO	WT/DEFL
Integral in eq (11)	YINT(IJ)	INT
ds	DS	
$\frac{dr}{d\theta}$	DRDT(IJ)	
$\frac{d^2r}{d\theta^2}$	SDRDT(IJ)	
$\frac{d\theta}{ds}$	DTDS	
$\frac{dr}{ds}$	DRDS	

```
@ RUN MITCHE,16650,02,60
@TI FOR MAIN1,MAIN1
      COMMON /INTEG/F(100),AREA
      DIMENSION YINT(91)
      DIMENSION PV(91)
      DIMENSION C1(20),C3(20),C5(20),C7(10),C8(10),XN(40),B(91),T(91),
     1 R(91),DRDT(91),SDRDT(91),RHO(91),G (91),PM(91),PN(91),
     2        C2(20),THETA(91)
 44   READ(5, 10) N1,N3,N5,N7,N8,NN,IGO
 10   FORMAT (20I4)
C  IGO.NE.0 MEANS ANOTHER SET OF DATA FOLLOWS THIS SET.
      READ(5 ,20)(C1(I),I=1,N1)
      READ(5 ,20)(C3(I),I=1,N3)
      READ(5 ,20)(C5(I),I=1,N5)
      READ(5 ,20)(C7(I),I=1,N7)
      READ(5 ,20)(C8(I),I=1,N8)
      READ(5 ,20)(XN(I),I=1,NN)
 20   FORMAT (10E8.0)
      READ(5 ,30) XNHU,XLLAM,S,E,GAMMA
 30   FORMAT (5E15.0)
C INSERT INITIAL WRITE CARDS HERE.
C PRINT OUT ALL INPUT DATA AT START OF RUN. PROGRAM 1.
      WRITE(6,1000) N1,N3,N5,N7,N8,NN,IGO
 1000 FORMAT (19H1INTEGER PARAMETERS //(8I5))
      WRITE(6,1010)(C1(I),I=1,N1)
      WRITE(6,1020)(C3(I),I=1,N3)
      WRITE(6,1030)(C5(I),I=1,N5)
      WRITE(6,1040)(C7(I),I=1,N7)
      WRITE(6,1050)(C8(I),I=1,N8)
      WRITE(6,1060)(XN(I),I=1,NN)
 1010 FORMAT ( 3H0C1 //(8E15.5))
 1020 FORMAT (3H0C3  //(8E15.5))
 1030 FORMAT (3H0C5  //(8E15.5))
 1040 FORMAT (3H0C7  //(8E15.5))
 1050 FORMAT (3H0C8  //(8E15.5))
 1060 FORMAT (2H0N   //(8E15.5))
      WRITE (6,1070)
 1070 FORMAT(1H0 11X,3H NU,15X,7H LAMBDA,18X,1HS,19X,1HE,14X,5HGAMMA )
      WRITE( 6,1080) XNHU,XLLAM,S,E,GAMMA
 1080 FORMAT (1H0,5E20.8)
      JCOUNT=0
      PI =3.141592654
      DO 1 I=1,N1
      C2(I)=2.0-C1(I)
      DO 1 J=1,N3
      DO 1 K=1,N5
      DO 1 L=1,N7
      DO 1 M=1,N8
      DO 1 N=1,NN
      A=C1(I)
```

```
      D= C2(I)
C JI GOES FROM 0 TO 90.
      DO 2 IJ =1,91
      JI= IJ-1
      THETA(IJ)=FLOAT(JI)*PI/180.0
      B(IJ)= C3(J)*(1.-C7(L)*COS(2.*THETA(IJ)))
      T(IJ)= C5(K)*(1.-C8(M)*COS(2.*THETA(IJ)))
      TERM = A     **XN(N)*SIN(THETA(IJ))**XN(N)+D     **XN(N)*
     1    COS(THETA(IJ))**XN(N)
      R(IJ)= A*D          /TERM**(1./XN(N))
      TP1=  D     **XN(N)*COS(THETA(IJ))**(XN(N)-1.0)*SIN(THETA(IJ))
      TP2 = A     **XN(N)*SIN(THETA(IJ))**(XN(N)-1.0)*COS(THETA(IJ))
      DRDT(IJ)= R(IJ)*(TP1-TP2)/TERM
      IF(XN(N).EQ.2.0.AND.THETA(IJ).EQ.0.0) GO TO 121
      TP3 =     A ** XN(N)*
     1             (SIN(THETA(IJ))**XN(N)-(XN(N)-1.0)*SIN(THETA(IJ))**
     2 (XN(N)-2.) * COS(THETA(IJ))**2 )
      IF(XN(N).EQ.2.0.AND.IJ.EQ.91) GO TO 122
      GO TO 125
 121  TP3 = A**2*(-1.0)
 125  CONTINUE
      TP4=     D ** XN(N)*
     1   (COS(THETA(IJ))**XN(N)-(XN(N)-1.0)*COS(THETA(IJ))**
     2 (XN(N)-2.)* SIN(THETA(IJ))**2 )
      GO TO 123
 122  TP4 = D**2 * (-1.0)
 123  CONTINUE
      SDRDT(IJ)=      A*D    *(XN(N)+1.0)*(TP2-TP1)*TERM**
     1 (-2.-(1./XN(N)))*(TP2-TP1)+ A*D*TERM**(-1.-(1./XN(N)))*(TP3+TP4)
      RHO(IJ)=(R(IJ)**2+2.0*DRDT(IJ)**2-R(IJ)*SDRDT(IJ))/
     1   (R(IJ)**2+DRDT(IJ)**2)**(3./2.)
C RHO  IS  NOW  K.
      IF(T(IJ)*RHO(IJ).GE.2.0) GO TO 1
      DO 1971 JMA=1,11
      Y=-T(IJ)*.5 + (T(IJ)*.5) * (FLOAT(JMA-1)/5.0 )
 1971 F(JMA)=Y**2/(1.0+RHO(IJ)*Y)
      CALL YSIMP(10.0/T(IJ))
      YINT(IJ)=AREA
      DIV=  RHO(IJ)/T(IJ)
      DO 1972 JMA=1,11
      Y=-T(IJ)*.5 + (T(IJ)*.5) * (FLOAT(JMA-1)/5.0 )
      SUM = (1.+ RHO(IJ)* Y)
1972  F(JMA)=DIV**2+2.0*DIV*Y/(SUM*YINT(IJ))+Y**2/(SUM*YINT(IJ))**2
      CALL YSIMP(10.0/T(IJ))
      G(IJ)= AREA
      DTDS = 1./ SQRT(R(IJ)**2+DRDT(IJ)**2)
      DRDS = DRDT(IJ)*DTDS
      PV(IJ)= R(IJ)*SIN(THETA(IJ))*DTDS-COS(THETA(IJ))*DRDS
      PN(IJ)= SIN(THETA(IJ))*DRDS+R(IJ)*COS(THETA(IJ))*DTDS
   2  CONTINUE
```

```
C NOW WE WANT MPRIME(CALLED PM).
C WE  DO SOME SIMPSON INTEGRATION.
      DO 101 JJ=1,91
      F(JJ)= R(JJ)*COS(THETA(JJ))*SQRT(R(JJ)**2+DRDT(JJ)**2)*G(JJ)/B(JJ)
     1   *(1.0-XNHU**2)
 101  CONTINUE
      CALL SIMP(1.0)
      PAR1 =AREA
      DO 102 JJ=1,91
      F(JJ)= SIN(THETA(JJ))*DRDT(JJ)* RHO(JJ)/(B(JJ)*T(JJ))
 102  CONTINUE
      CALL SIMP(1.0)
      PAR2 =AREA
      DO 103  JJ=1,91
      F(JJ)= COS(THETA(JJ))*R(JJ) * RHO(JJ)/(B(JJ)*T(JJ))
 103  CONTINUE
      CALL SIMP(1.0)
      PAR3= AREA
      DO 104 JJ= 1,91
      F(JJ)= SQRT(R(JJ)**2 +DRDT(JJ)**2) *G(JJ)/B(JJ)
     1   *(1.0-XNHU**2)
 104  CONTINUE
      CALL SIMP(1.0)
      PAR4 = AREA
C NOW INTEGRATE TO GET  W SUB LAMBDA.
      DO 105 JJ=1,91
      F(JJ)= SQRT(R(JJ)**2+DRDT(JJ)**2)*B(JJ)*T(JJ)*PI/180.
 105  CONTINUE
      CALL SIMP(1.0)
      PAR5 = AREA
C USE PAR5 TO COMPUTE W SUB LAMBDA LATER ON.
C  NOW GET PM.
      DO 106 JJ=1,91
      PM(JJ)= (PAR1-PAR2-PAR3)/PAR4 -R(JJ)* COS(THETA(JJ))
 106  CONTINUE
C FORM NEW LOOP TO GET BMIN.
      JMAX=0
      DMAX =0.0
      DO 8811 IJ =1,91
      DIV = PM(IJ)*RHO(IJ)/(B(IJ)*T(IJ))
      DENOM = DIV- PM(IJ)*T(IJ)/(B(IJ)*(2.-RHO(IJ)*T(IJ))*YINT(IJ))
     1    + PN(IJ)/(B(IJ)* T(IJ))
      IF (ABS(DENOM).GT.DMAX)JMAX=IJ-1
      DMAX =  AMAX1 (ABS(DENOM), ABS(DMAX))
 8811 CONTINUE
      BMIN = 2.0/DMAX
      XLAM = XLLAM /(BMIN*S)
      WLAM = 4.*XLAM**2*GAMMA*PAR5
C NOW COMPUTE SOME VALUES TO COMPARE WITH CONSTAINTS.
      TCON4 = XLAM*C5(K)*(1.+ABS(C8(M)))
```

```
      TCON5 = XLAM*(2.*C1(I)+T(1))
      TCON6 = XLAM*(2.*C2(I)+T(91))
      TCON7 =       C3(J)*(1.+ABS(C7(L)))
      TCON8 = XLAM*(2.*C2(I)-T(91))
C NOW GO TO DOOPS IF CONSTRAINTS ARE NOT SATISFIED.
      DO 107 JJ=1,91
      TM1 = (1.-XNHU**2)*G(JJ)*T(JJ)*PM(JJ)*R(JJ)*COS(THETA(JJ))
      TM2 =              RHO(JJ)*PM(JJ)*PN(JJ)
      DS= SQRT(R(JJ)**2+DRDT(JJ)**2)
      TM3 =              RHO(JJ)*PN(JJ)*R(JJ)*COS(THETA(JJ))
      F(JJ)=1./(B(JJ)*T(JJ))*(PN(JJ)**2+12./5.*(1.+XNHU)*
     1 PV(JJ)**2-TM1+TM2-TM3)*DS  *PI/180.
  107 CONTINUE
      CALL SIMP(1.0)
      ASTAR=AREA
      DELAM=XLAM*ASTAR*BMIN*S/E
      RATIO=WLAM/DELAM
C COMPUTE SOME OUTPUT ITEMS.
      IF(MOD(JCOUNT,25).NE.0) GO TO 7
      WRITE(6,4)
    4 FORMAT(1H1 1X 'WT/DEFL'4X'WEIGHT'6X'G1'6X'G2'5X'G3'6X'G4'6X'G5'
     1  7X'G6'3X'LAMBDA'4X'A'7X'C'8X'D'7X'E'7X'F'7X'N'2X'TM'//)
    7 WRITE(6,5)RATIO,WLAM,DELAM,TCON4,TCON5,TCON6,TCON7,TCON8,XLAM,
     1  C1(I),C3(J),C5(K),C7(L),C8(M),XN(N),JMAX
    5 FORMAT(1H0,2F9.2,F9.3,12F8.3,I3)
      JCOUNT=JCOUNT+1
    1 CONTINUE
      IF(IGO.NE.0) GO TO 44
C WE HAVE NOW FINISHED MAJOR LOOP.
      STOP
      END
@TI FOR SINT,SINT
      SUBROUTINE  SIMP(HHH)
      COMMON /INTEG /F(100),AREA
      H = 1./(3.*HHH)
      AREA=0.0
      ODD=0.0
      EVEN=0.0
      DO 21 I =2,90,2
   21 EVEN =EVEN + F(I)
      DO 22 I =3,89,2
   22 ODD =ODD+ F(I)
      AREA = H* (F(1)+F(91)+4.*EVEN+ 2.*ODD)
      RETURN
      END
@TI FOR YSUB,YSUB
      SUBROUTINE YSIMP(HHH)
      COMMON /INTEG /F(100),AREA
      H = 1./(3.*HHH)
      AREA=0.0
```

```
      ODD=0.0
      EVEN=0.0
      DO 21 I=2,10,2
 21   EVEN =EVEN + F(I)
      DO 22 I=3,9,2
 22   ODD =ODD+ F(I)
      AREA = H*(F(1)+F(11)+4.0*EVEN+2.0*ODD)
      RETURN
      END
@ XQT MAIN1
   1    5    1    6    3    1    0    0
1.21
12.7     13.      13.3     13.6     14.
.305
-.025    0.       .025     .05      .075     .1
.4       .45      .5
6.
.30              1000000.        150000.          30000000.        .290
@ EOF
@ FIN
```

```
@ RUN MITCHE,16650,02,60
@IT    FOR  MAIN2,MAIN2
       COMMON  /INTEG/F(100),AREA
       DIMENSION XLN(91),XLV(91),XLM(91)
       DIMENSION YINT(91)
       DIMENSION PV(91)
       DIMENSION G(91)
       DIMENSION B(91),T(91),R(91),DRDT(91),SDRDT(91),RHO(91),
      1  PM(91),PN(91),SIGTL(91),THETA(91)
 10    READ(5,20) C1,C3,C5,C7,C8,XN
 20    FORMAT (10E8.0)
       IF (C1.EQ.9999.) STOP
       READ(5,30) XNHU,XLLAM,S,E,GAMMA
 30    FORMAT (5E15.0)
C  INSERT  INITIAL  PRINT  HERE.
C PRINT OUT ALL INPUT DATA AT START OF RUN. PROGRAM 3.
       WRITE(6,1000) C1,C3,C5,C7,C8,XN
 1000 FORMAT (  1H1, 5X,2HC1,8X,2HC3,8X,2HC5,8X,2HC7,8X,2HC8,8X,1HN,//
      1    6F10.3)
       WRITE (6,1070)
       WRITE (6,1080) XNHU,XLLAM,S,E,GAMMA
C INSERT FORMATS 1070 AND 1080 HERE.
 1070 FORMAT(1H0 10X,3H NU,15X,7H LAMBDA,18X,1HS,19X,1HE,14X,5HGAMMA )
 1080 FORMAT (1H0,5E20.8)
       PI =3.141592654
       C2=2.0-C1
       C4 = C3*C7
       C6 = C5*C8
       A =C1
       D=C2
       DO 2  I =1,91
       JI = I-1
       THETA(I)=FLOAT(JI)*PI/180.
       B(I) = C3 * (1.-C7 *COS(2.*THETA(I)))
       T(I) = C5 * (1.-C8 *COS(2.*THETA(I)))
       TERM = A ** XN *SIN(THETA(I))**XN+D  **XN *
      1   COS(THETA(I))** XN
       R(I)=A*D/(TERM **(1./XN) )
       TP1 = D **XN* COS(THETA(I))**(XN-1.0)*SIN(THETA(I))
       TP2 = A **XN* SIN(THETA(I))**(XN-1.0)*COS(THETA(I))
       DRDT(I)=R(I) *(TP1-TP2)/TERM
       IF(XN.EQ.2.0.AND.I .EQ.1) GO TO 121
       TP3 = A **XN*(SIN(THETA(I))** XN-(XN-1.0)* SIN (THETA(I))**
      1    (XN-2.0)*COS (THETA(I))**2 )
       IF(XN.EQ.2.0.AND.I .EQ.91) GO TO 122
       GO TO 125
 121   TP3 = A**2*(-1.0)
 125   CONTINUE
       TP4 = D**XN *(COS(THETA(I))** XN-(XN-1.0)* COS (THETA(I))**
      1    (XN-2.0)*SIN (THETA(I))**2 )
```

```
      GO TO 123
  122 TP4 = D**2 * (-1.0)
  123 CONTINUE
      SDRDT(I)= A*D * (XN+1.0)* (TP2-TP1)* TERM**
     1 (-2.-(1./XN))*(TP2-TP1)+ A*D*TERM**(-1.-(1./XN))*(TP3+TP4)
      RHO(I) = (R(I)**2+2.*DRDT(I)**2-R(I)*SDRDT(I))
     1 / (R(I)**2+DRDT(I)**2)**(3.0/2.0)
      DO 1971 J=1,11
      Y=-T(I)*.5+ T(I)*.5*(FLOAT(J-1)/5.0)
 1971 F(J)=Y**2/(1.0+RHO(I)*Y)
      CALL YSIMP(10.0/T(I))
      YINT(I)=AREA
      DIV = RHO(I)/T(I)
      DO 1972 J=1,11
      Y=-T(I)*.5+ T(I)*.5*(FLOAT(J-1)/5.0)
      SUM=(1.0+RHO(I)*Y)
 1972 F(J)=DIV**2+2.*DIV*Y/(SUM*YINT(I))+Y**2/(SUM*YINT(I))**2
      CALL YSIMP(10.0/T(I))
      G(I) = AREA
      DTDS =1./SQRT(R(I)**2 +DRDT(I)**2)
      DRDS = DRDT(I)* DTDS
      PV(I)= R(I)*SIN(THETA(I))*DTDS-COS(THETA(I))*DRDS
      PN(I) = SIN(THETA(I))*DRDS +R(I)*COS(THETA(I)) *DTDS
    2 CONTINUE
C NOW WE INTEGRATE TO GET MPRIME (CALLED PM).
      DO 101   J =1,91
      F(J)=R(J)*COS(THETA(J))*SQRT(R(J)**2+DRDT(J)**2)
     1 *G(J)/B(J)*(1.0-XNHU**2)
  101 CONTINUE
      CALL SIMP(1.0)
      PAR1=AREA
      DO 102 J =1,91
      F(J)= SIN(THETA(J))*DRDT(J)*RHO(J)/(B(J)*T(J))
  102 CONTINUE
      CALL SIMP(1.0)
      PAR2 =AREA
      DO 103 J=1,91
      F(J)= R(J)*COS(THETA(J))*RHO(J)/(B(J)*T(J))
  103 CONTINUE
      CALL SIMP(1.0)
      PAR3= AREA
      DO 104 J=1,91
      F(J)=SQRT(R(J)**2+DRDT(J)**2) *G(J)/B(J)*(1.0-XNHU**2)
  104 CONTINUE
      CALL SIMP(1.0)
      PAR4 = AREA
C NOW WE INTEGRATE TO GET W SUB LAMBDA. USE PAR5 LATER ON.
      DO 105  J=1,91
      F(J)=SQRT (R(J)**2 +DRDT(J)**2)* B(J)*T(J)*PI/180.0
  105 CONTINUE
```

```
      CALL SIMP(1.0)
      PAR5 = AREA
C NOW GET PM.
      DO 106 J=1,91
      PM(J)= (PAR1-PAR2-PAR3)/PAR4 -R(J)*COS(THETA(J))
 106  CONTINUE
C FORM LOOP TO GET BMIN AND SIGTL.
      DMAX = 0.0
      DO 8811 I=1,91
      DIV = PM(I)* RHO(I)/(B(I)*T(I))
      SIGTL(I)=  DIV - PM(I)*T(I)/(B(I)*(2.-RHO(I)* T(I))*YINT(I))
     1  +PN(I)/(B(I)*T(I))
      DMAX = AMAX1(ABS(SIGTL(I)),ABS(DMAX))
 8811 CONTINUE
      BMIN = 2.0/DMAX
      XLAM= XLLAM /(BMIN*S )
      FACTOR=BMIN*S*.5
      DO 8833 I =1,91
      SIGTL(I)=FACTOR*SIGTL(I)
      XLN(I)=XLAM*FACTOR*PN(I)
      XLV(I)=XLAM*FACTOR*PV(I)
      XLM(I)=XLAM**2*FACTOR*PM(I)
 8833 CONTINUE
      DO 107 JJ=1,91
      TM1 = (1.-XNHU**2)*G(JJ)*T(JJ)*PM(JJ)*R(JJ)*COS(THETA(JJ))
      TM2 =                RHO(JJ)*PM(JJ)*PN(JJ)
      DS= SQRT(R(JJ)**2+DRDT(JJ)**2)
      TM3 =                RHO(JJ)*PN(JJ)*R(JJ)*COS(THETA(JJ))
      F(JJ)=1./(B(JJ)*T(JJ))*(PN(JJ)**2+12./5.*(1.+XNHU)*
     1 PV(JJ)**2-TM1+TM2-TM3)*DS  *PI/180.
 107  CONTINUE
      CALL SIMP(1.0)
      ASTAR=AREA
      DELAM=XLAM*ASTAR*BMIN*S/E
      WLAM = 4.* XLAM **2* GAMMA *PAR5
C NOW COMPUTE SOME OUTPUT ITEMS.
      TCON4=XLAM*C5*(1.+ABS(C8))
      TCON5= XLAM * (2.* A +T(1))
      TCON6= XLAM * (2.* D +T(91))
      TCON7=          C3 * (1.+ ABS(C7))
      TCON8=XLAM*(2.0*C2-T(91))
C NOW COMPUTE STRESS.FOR EACH THETA.
C NOW WE PRINT OUT DESIRED OUTPUT, THEN READ ANOTHER DATA SET.
      WRITE(6,40)
 40   FORMAT(1H0 1X 'WT/DEFL'4X'WEIGHT'6X'G1'6X'G2'5X'G3'6X'G4'6X'G5'
     1  7X'G6'3X'LAMDA'5X'A'7X'C'8X'D'7X'E'7X'F'7X'N'/ )
      RATIO=WLAM/DELAM
      WRITE(6,70)RATIO,WLAM,DELAM,TCON4,TCON5,TCON6,TCON7,TCON8,XLAM,
     1   C1,C3,C5,C7,C8,XN
 70   FORMAT(1H0,2F9.2,F9.3,12F8.3)
```

```
      WRITE (6,80)
 80   FORMAT(1H0'THETA'6X'R'9X'LAMR'9X'T'8X'LAMT'9X'L'9X'K'9X'INT'
     1  7X'LAMN'7X'LAMV'4X'LAM**2*M'6X'SIGMA'//)
      DO  64   I =1,91
      J =I-1
      RLAM=R(I)*XLAM
      TLAM=T(I)*XLAM
      WRITE(6,90)J,R(I),RLAM,T(I),TLAM,B(I),RHO(I),YINT(I),XLN(I),
     1  XLV(I),XLM(I),SIGTL(I)
 90   FORMAT(1H  I5,11E11.4)
      IF(MOD(J,5).EQ.0.AND.J.NE.0) WRITE(6,110)
 64   CONTINUE
 110  FORMAT (1H )
      GO TO 10
      STOP
      END
@N FOR SINT,SINT
      SUBROUTINE SIMP(HHH)
      COMMON /INTEG /F(100),AREA
      H = 1.0/(3.0*HHH)
      AREA=0.0
      ODD=0.0
      EVEN=0.0
      DO 21 I =2,90,2
 21   EVEN =EVEN + F(I)
      DO 22 I =3,89,2
 22   ODD =ODD+ F(I)
      AREA = H* (F(1)+F(91)+4.*EVEN+ 2.*ODD)
      RETURN.
      END
@N FOR YSUB,YSUB
      SUBROUTINE YSIMP(HHH)
      COMMON /INTEG /F(100),AREA
      H = 1./(3.*HHH)
      AREA=0.0
      ODD=0.0
      EVEN=0.0
      DO 21 I =2,10,2
 21   EVEN =EVEN + F(I)
      DO 22 I =3, 9,2
 22   ODD =ODD+ F(I)
      AREA = H* (F(1)+F(11)+4.*EVEN+ 2.*ODD)
      RETURN
      END
@ XQT MAIN2
1.22    13.8     .305     0.       .45      5.5
.3               1000000.          150000.           30000000.        .29
1.      14.      .248     0.       0.       2.
.3               1000000.          150000.           30000000.        .29
@ EOF
```

JRL. RES. SEC. C

(Paper 72C2–273)